

x86 Programming II

CSE 351 Autumn 2016

Instructor:

Justin Hsia

Teaching Assistants:

Chris Ma

Hunter Zahn

John Kaltenbach

Kevin Bi

Sachin Mehta


Suraj Bhat

Thomas Neuman

Waylon Huang

Xi Liu

Yufang Sun



<http://xkcd.com/1652/>

Administrivia


- ❖ Lab 2 released tomorrow
 - Learn to use gdb and look at assembly code
- ❖ Homework 1 due on Friday (10/21)

Address Computation Instruction

- ❖ `leaq src, dst`
 - “lea” stands for *load effective address*
 - `src` is address expression (any of the formats we’ve seen)
 - `dst` is a register
 - Sets `dst` to the *address* computed by the `src` expression
(**does not go to memory! – it just does math**)
 - Example: `leaq (%rdx, %rcx, 4), %rax`

$$\begin{matrix} D & Rb & Ri & S \\ \text{movq} (\%rdx, \%rcx, 4), \%rax \end{matrix} \rightarrow p = x[i]$$
- ❖ **Uses:**
 - Computing addresses **without** a memory reference
 - e.g., translation of `p = &x[i];` *lea (address-of)*
 - Computing arithmetic expressions of the form $x + k * i$
 - Though `k` can only be 1, 2, 4, or 8 \overline{S}
↖

Example: lea vs. mov



Registers

%rax	0x110
%rbx	0x8
%rcx	0x4
%rdx	0x100
%rdi	0x100
%rsi	0x1

Memory Word Address

0x400	0x120
0xF	0x118
0x8	0x110
0x10	0x108
0x1	0x100

$0x100 + 0x4 * 4 + 0 = 0x110$
 $Reg[Rb] + Reg[Ri] * S + D$

leaq	^D (^{Rb} %rdx, ^{Ri} %rcx, ^S 4),	%rax	→ 0x110 (addr)
movq	(%rdx, %rcx, 4),	%rbx	→ 0x8 (data)
leaq	(^{Rb} %rdx),	%rdi	→ 0x100
movq	(%rdx),	%rsi	→ 0x1

Example: lea vs. mov (solution)

Registers		Memory	Word Address
%rax	0x110	0x400	0x120
%rbx	0x8	0xF	0x118
%rcx	0x4	0x8	0x110
%rdx	0x100	0x10	0x108
%rdi	0x100	0x1	0x100
%rsi	0x1		

```
leaq (%rdx,%rcx,4), %rax
movq (%rdx,%rcx,4), %rbx
leaq (%rdx), %rdi
movq (%rdx), %rsi
```

Arithmetic Example

```

long arith(long x, long y, long z)
{
    long t1 = x + y;
    long t2 = z + t1;
    long t3 = x + 4;
    long t4 = y * 48;
    long t5 = t3 + t4;
    long rval = t2 * t5;
    return rval;
}
    
```

Register	Use(s)
%rdi	1 st argument (x)
%rsi	2 nd argument (y)
%rdx	3 rd argument (z)

$$addr = Reg[Rb] + Reg[Ri] * S + D$$

$$y + y * 2 + 0 = 3y$$

```

arith:
    leaq    (%rdi,%rsi), %rax → x+y
    addq   %rdx, %rax → x+y+z
    leaq   (%rsi,%rsi,2), %rdx → 3y
    salq   $4, %rdx → 16*(3y) = 48y
    leaq   4(%rdi,%rdx), %rcx → x+48y+4
    imulq  %rcx, %rax → (x+y+z)(x+48y+4)
    ret
    
```

❖ Interesting Instructions

- leaq: “address” computation
- salq: shift
- imulq: multiplication
 - Only used once!

Arithmetic Example

```

long arith(long x, long y, long z)
{
    long t1 = x + y;
    long t2 = z + t1;
    long t3 = x + 4;
    long t4 = y * 48;
    long t5 = t3 + t4;
    long rval = t2 * t5;
    return rval;
}

```

Register	Use(s)
%rdi	x
%rsi	y
%rdx	z, t4
<u>%rax</u>	<u>t1, t2, rval</u>
%rcx	t5

```

arith:
    leaq    (%rdi,%rsi), %rax    # rax/t1 = x + y
    addq   %rdx, %rax          # rax/t2 = t1 + z
    leaq   (%rsi,%rsi,2), %rdx  # rdx    = 3 * y
    salq   $4, %rdx           # rdx/t4 = (3*y) * 16
    leaq   4(%rdi,%rdx), %rcx   # rcx/t5 = x + t4 + 4
    imulq  %rcx, %rax          # rax/rval = t5 * t2
    ret

```

x86 Control Flow

- ❖ Condition codes
- ❖ Conditional and unconditional branches
- ❖ Loops
- ❖ Switches

Control Flow

Register	Use(s)
%rdi	1 st argument (x)
%rsi	2 nd argument (y)
%rax	return value

```
long max(long x, long y)
{
    long max;
    if (x > y) {
        max = x;
    } else {
        max = y;
    }
    return max;
}
```

```
max:
    ???
    movq   %rdi, %rax
    ???
    ???
    movq   %rsi, %rax
    ???
    ret
```

Control Flow

Register	Use(s)
%rdi	1 st argument (x)
%rsi	2 nd argument (y)
%rax	return value

```

long max(long x, long y)
{
    long max;
    if (x > y) {
        max = x;
    } else {
        max = y;
    }
    return max;
}
    
```

Conditional jump

Unconditional jump

```

max:
    if TRUE
    if x <= y then jump to else
    if FALSE
    movq %rdi, %rax
    jump to done
else:
    movq %rsi, %rax
done:
    ret
    
```

Conditionals and Control Flow

- ❖ Conditional branch/*jump*
 - Jump to somewhere else if some *condition* is true, otherwise execute next instruction
- ❖ Unconditional branch/*jump*
 - *Always* jump when you get to this instruction
- ❖ Together, they can implement most control flow constructs in high-level languages:
 - `if (condition) then {...} else {...}`
 - `while (condition) {...}`
 - `do {...} while (condition)`
 - `for (initialization; condition; iterative) {...}`
 - `switch {...}`

Jumping

- ❖ j^* Instructions
 - Jumps to **target** (argument – actually just an address)
 - Conditional jump relies on special *condition code registers*

Instruction	Condition	Description
<code>jmp target</code>	1	Unconditional
<code>je target</code>	ZF	Equal / Zero
<code>jne target</code>	$\sim ZF$	Not Equal / Not Zero
<code>js target</code>	SF	Negative
<code>jns target</code>	$\sim SF$	Nonnegative
<code>jg target</code>	$\sim (SF \wedge OF) \wedge \sim ZF$	Greater (Signed)
<code>jge target</code>	$\sim (SF \wedge OF)$	Greater or Equal (Signed)
<code>jl target</code>	$(SF \wedge OF)$	Less (Signed)
<code>jle target</code>	$(SF \wedge OF) \mid ZF$	Less or Equal (Signed)
<code>ja target</code>	$\sim CF \wedge \sim ZF$	Above (unsigned)
<code>jb target</code>	CF	Below (unsigned)


← special (no condition!)

don't sweat the details



Processor State (x86-64, partial)

- ❖ Information about currently executing program
 - Temporary data (`%rax`, ...)
 - Location of runtime stack (`%rsp`)
 - Location of current code control point (`%rip`, ...)
 - Status of recent tests (`CF`, `ZF`, `SF`, `OF`)
 - Single bit registers:

Registers




current top of the Stack



Condition Codes (Implicit Setting)

- ❖ *Implicitly* set by **arithmetic** operations
 - (think of it as side effects)
 - Example: `addq src, dst` \leftrightarrow `t = a+b`
 - **CF=1** if carry out from MSB (unsigned overflow)
 - **ZF=1** if `t==0`
 - **SF=1** if `t<0` (assuming signed, actually just if MSB is 1)
 - **OF=1** if two's complement (signed) overflow
(`a>0 && b>0 && t<0`) || (`a<0 && b<0 && t>=0`)
 - **Not set by `leaq` instruction (beware!)**



Condition Codes (Explicit Setting: Compare)

❖ Explicitly set by **Compare** instruction

■ `cmpq src2, src1`

■ `cmpq b, a` sets flags based on $a-b$, but doesn't store

■ **CF=1** if carry out from MSB (used for unsigned comparison)


■ **ZF=1** if $a==b$

■ **SF=1** if $(a-b) < 0$ (signed)

■ **OF=1** if two's complement (signed) overflow

$(a > 0 \ \&\& \ b < 0 \ \&\& \ (a-b) < 0) \ ||$

$(a < 0 \ \&\& \ b > 0 \ \&\& \ (a-b) > 0)$



Condition Codes (Explicit Setting: Test)

❖ *Explicitly* set by **Test** instruction

- `testq src2, src1`
- `testq b, a` sets flags based on `a&b`, but doesn't store
 - Useful to have one of the operands be a *mask*

■ Can't have carry out (**CF**) or overflow (**OF**)

■ **ZF=1** if `a&b==0`

ZF=1: if $a \& a == 0 \rightarrow a == 0$


■ **SF=1** if `a&b<0` (signed)

SF=1: if $a \& a < 0 \rightarrow a < 0$

■ Example: `testq %rax, %rax`

- Tells you if (+), 0, or (-) based on ZF and SF

ZF	SF	a
0	0	$a > 0$
0	1	$a < 0$
1	0	$a == 0$
1	1	not possible



Reading Condition Codes

❖ set* Instructions

stores logical result of condition (1=True, 0=False)

- Set a low-order byte to 0 or 1 based on condition codes
- Does not alter remaining 7 bytes

same instruction endings as j* and same condition flag statements

Instruction	Condition	Description
<u>sete</u> dst	ZF	Equal / Zero
<u>setne</u> dst	~ZF	Not Equal / Not Zero
<u>sets</u> dst	SF	Negative
<u>setns</u> dst	~SF	Nonnegative
<u>setg</u> dst	~(SF^OF) & ~ZF	Greater (Signed)
<u>setge</u> dst	~(SF^OF)	Greater or Equal (Signed)
<u>setl</u> dst	(SF^OF)	Less (Signed)
<u>setle</u> dst	(SF^OF) ZF	Less or Equal (Signed)
<u>seta</u> dst	~CF & ~ZF	Above (unsigned ">")
<u>setb</u> dst	CF	Below (unsigned "<")

x86-64 Integer Registers

❖ Accessing the low-order byte:

l for "low-order byte"

"b" for byte

<code>%rax</code>	<code>%al</code>
<code>%rbx</code>	<code>%bl</code>
<code>%rcx</code>	<code>%cl</code>
<code>%rdx</code>	<code>%dl</code>
<code>%rsi</code>	<code>%sil</code>
<code>%rdi</code>	<code>%dil</code>
<code>%rsp</code>	<code>%spl</code>
<code>%rbp</code>	<code>%bpl</code>

8 bits

<code>%r8</code>	<code>%r8b</code>
<code>%r9</code>	<code>%r9b</code>
<code>%r10</code>	<code>%r10b</code>
<code>%r11</code>	<code>%r11b</code>
<code>%r12</code>	<code>%r12b</code>
<code>%r13</code>	<code>%r13b</code>
<code>%r14</code>	<code>%r14b</code>
<code>%r15</code>	<code>%r15b</code>

8 bits

Reading Condition Codes

Register	Use(s)
%rdi	1 st argument (x)
%rsi	2 nd argument (y)
%rax	return value

❖ set* Instructions

- Set a low-order byte to 0 or 1 based on condition codes
- Operand is byte register (e.g. al, dl) or a byte in memory
- Do not alter remaining bytes in register
 - Typically use movzbl (zero-extended mov) to finish job

```
int gt(long x, long y)
{
    return x > y;
}
```

```

{
  cmpq   %rsi, %rdi   # x - y
  setg   %al          # ?_al = (x > y)
  movzbl %al, %eax    # %rax = (x > y)
  ret
}

```

Handwritten notes:

- Red 'y' above %rsi, red 'x' above %rdi
- Red bracket on the left of the assembly block
- Red text: *greater than* under the `setg` instruction
- Red text: *%rax = (x > y)* next to the `movzbl` instruction

Reading Condition Codes

Register	Use(s)
%rdi	1 st argument (x)
%rsi	2 nd argument (y)
%rax	return value

❖ set* Instructions

- Set a low-order byte to 0 or 1 based on condition codes
- Operand is byte register (e.g. al, dl) or a byte in memory
- Do not alter remaining bytes in register
 - Typically use movzbl (zero-extended mov) to finish job

```
int gt(long x, long y)
{
    return x > y;
}
```

```
cmpq    %rsi, %rdi    # Compare x:y
setg    %al           # Set when >
movzbl  %al, %eax     # Zero rest of %rax
ret
```

Aside: movz and movs

movz__ src, regDest

Move with zero extension

movs__ src, regDest

Move with sign extension

- Copy from a *smaller* source value to a *larger* destination
- Source can be memory or register; Destination *must* be a register
- Fill remaining bits of dest with **zero** (movz) or **sign bit** (movs)


movzSD / movsSD:

S – size of source (**b** = 1 byte, **w** = 2)

D – size of dest (**w** = 2 bytes, **l** = 4, **q** = 8)

Example: `movzq %al, %rbx`

1 byte (pointing to %al)
8 bytes (pointing to %rbx)



Aside: movz and movs

movz__ src, regDest

Move with zero extension

movs__ src, regDest

Move with sign extension

- Copy from a *smaller* source value to a *larger* destination
- Source can be memory or register; Destination *must* be a register
- Fill remaining bits of dest with **zero** (movz) or **sign bit** (movs)

movzSD / movsSD:


S – size of source (**b** = 1 byte, **w** = 2)

D – size of dest (**w** = 2 bytes, **l** = 4, **q** = 8)

Note: In x86-64, any instruction that generates a 32-bit (long word) value for a register also sets the high-order portion of the register to 0. Good example on p. 184 in the textbook.

Example:
 movsbl (%rax), %ebx

Copy 1 byte from memory into 8-byte register & sign extend it



Choosing instructions for conditionals

replace "j" with "set" to get other instructions

		<code>cmp b,a</code>	<code>test a,b</code>
<code>je</code>	"Equal"	<code>a == b</code>	<code>a&b == 0</code>
<code>jne</code>	"Not equal"	<code>a != b</code>	<code>a&b != 0</code>
<code>js</code>	"Sign" (negative)		<code>a&b < 0</code>
<code>jns</code>	(non-negative)		<code>a&b >= 0</code>
<code>jg</code>	"Greater"	<code>a > b</code>	<code>a&b > 0</code>
<code>jge</code>	"Greater or equal"	<code>a >= b</code>	<code>a&b >= 0</code>
<code>jl</code>	"Less"	<code>a < b</code>	<code>a&b < 0</code>
<code>jle</code>	"Less or equal"	<code>a <= b</code>	<code>a&b <= 0</code>
<code>ja</code>	"Above" (unsigned >)	<code>a > b</code>	
<code>jb</code>	"Below" (unsigned <)	<code>a < b</code>	

```

cmp 5,(p)
je:  *p == 5
jne: *p != 5
jg:  *p > 5
jl:  *p < 5
    
```

```

test a,a
je:  a == 0
jne: a != 0
jg:  a > 0
jl:  a < 0
    
```

```

test a,0x1
je:  a_LSB == 0
jne: a_LSB == 1
    
```

Typically come in pairs:

- ① test or compare
- ② jump or set

Choosing instructions for conditionals

		<u>cmp b,a</u>	test a,b
je	"Equal"	a == b	a&b == 0
jne	"Not equal"	a != b	a&b != 0
js	"Sign" (negative)		a&b < 0
jns	(non-negative)		a&b >= 0
jg	"Greater"	a > b	a&b > 0
<u>jge</u>	"Greater or equal"	<u>a >= b</u>	a&b >= 0
j1	"Less"	a < b	a&b < 0
jle	"Less or equal"	a <= b	a&b <= 0
ja	"Above" (unsigned >)	a > b	
jb	"Below" (unsigned <)	a < b	

Register	Use(s)
%rdi	argument <u>x</u>
%rsi	argument y
%rax	return value

```
if (x < 3) {
    return 1;
}
return 2;
```

```
cmpq $3, %rdi
jge T2
T1: # x < 3:
    movq $1, %rax
    ret
T2: # !(x < 3): x ≥ 3
    movq $2, %rax
    ret
```

cmpq \$3, %rdi
jge T2
T2 ← *label (addr)*

Your Turn!

Register	Use(s)
%rdi	1 st argument (x)
%rsi	2 nd argument (y)
%rax	return value

```

long absdiff(long x, long y)
{
    long result;
    if (x > y)
        result = x-y;
    else
        result = y-x;
    return result;
}

```

```

absdiff:
        cmpq    %rsi, %rdi
        jle    .L4
                                     # x > y:
    movq    %rdi, %rax
    subq    %rsi, %rax
    ret

.L4:                                     # x <= y:
    movq    %rsi, %rax
    subq    %rdi, %rax
    ret

```

❖ Can view in provided `control.s`

■ `gcc -Og -S -fno-if-conversion control.c`

Your Turn! (solution)

Register	Use(s)
%rdi	1 st argument (x)
%rsi	2 nd argument (y)
%rax	return value

```
long absdiff(long x, long y)
{
    long result;
    if (x > y)
        result = x-y;
    else
        result = y-x;
    return result;
}
```

```
absdiff:
    cmpq    %rsi, %rdi    # x:y
    jle    .L4
                                # x > y:
    movq    %rdi, %rax
    subq    %rsi, %rax
    ret
.L4:                                # x <= y:
    movq    %rsi, %rax
    subq    %rdi, %rax
    ret
```

❖ Can view in provided control.s

■ gcc -Og -S -fno-if-conversion control.c

Choosing instructions for conditionals

		cmp b,a	test a,b
je	"Equal"	② <u>a == b</u>	③ <u>a&b == 0</u>
jne	"Not equal"	a != b	a&b != 0
js	"Sign" (negative)		a&b < 0
jns	(non-negative)		a&b >= 0
jg	"Greater"	a > b	a&b > 0
jge	"Greater or equal"	a >= b	a&b >= 0
jl	"Less"	① <u>a < b</u>	a&b < 0
jle	"Less or equal"	a <= b	a&b <= 0
ja	"Above" (unsigned >)	a > b	
jb	"Below" (unsigned <)	a < b	

%al & %bl == 0 when either %al or %bl is false
↑ this is the else case!

```
if (x < 3 && x == y) {
    return 1;
} else {
    return 2;
}
```

```
① cmpq $3, %rdi } %al = (x < 3)
   setl %al
② cmpq %rsi, %rdi } %bl = (x == y)
   sete %bl
③ testb %al, %bl
   je T2 ← jump to T2 if (%al & %bl) == 0
T1: # x < 3 && x == y:
    movq $1, %rax
    ret
T2: # else
    movq $2, %rax
    ret
```

Summary

- ❖ `lea` is address calculation instruction
 - Does NOT actually go to memory
 - Used to compute addresses or some arithmetic expressions
- ❖ Control flow in x86 determined by status of Condition Codes
 - Showed **C**arry, **Z**ero, **S**ign, and **O**verflow, though others exist
 - Set flags with arithmetic instructions (implicit) or Compare and Test (explicit)
 - Set instructions read out flag values
 - Jump instructions use flag values to determine next instruction to execute